
It all started in London, UK, early 1996. During his advanced seminar about 100% generation with Designer 2000, Scott Hollows asked “What’s the ultimate purpose of the applications that we all design?”. After some debate he said: ”The ultimate purpose is to provide information to the users.” This principle is absolutely fundamental to information systems applications. It is simple, yet easy to forget. It highlights the importance of queries. This article illustrates a very useful technique to contribute to reaching this objective.

Background

We are all familiar with the term “drilling down”. It started to get popular with the advent of Executive Information Systems. Users may drill down by going from a contract shipping summary to individual shipments. They may then want to explore individual containers, then cases within containers, then items within cases. 

But there is more to drilling than drilling down. Users also want to be able to drill up. Let us say that the consignee of one of our shipments calls about a missing item. Users will probably retrieve the item details and check the status of the problematic item. If nothing looks abnormal, they may then want to check the items that were shipped in the same case as our missing item. Should these people have to make a note of the case reference, look up the menu for the case contents browse form, open it, enter query, key the case number in, execute query and contemplate the fruit of their labor? Of course not.

Drilling across is also very important. From a purchase order header, users often need to lookup customer details. From the purchase order lines, they also often want to be only one mouse click away from product details. No menu, no manual querying.

So this is what we want to achieve. From an order header/order line screen, the user may call a “Customer Details” form clicking a simple press button. The called form must be aware of what customer (if any) placed the order in the calling form. The  called form must auto query itself and bring up this customer’s details. Pretty staightforward, right? The same form can be called from a menu option too, with or without context. We must design this form as an object that can be called from anywhere in any application. We must not have to worry about dirty global variables. Modularity and Re-usability are the characteristics of good software engineering.

Yes, you can provide all this 100% generated. We’ll detail a simple method using a PL/SQL procedure in our template: std_call_form. 

We’ll first show you the design of the elements of our demonstration, i.e. the two forms: calling and called. Then we’ll show you the result when we run these two forms. You will notice that one  can generate 100% layouts that are perfectly acceptable. We’ll then explain exactly how we achieved those results. We will conclude with a quick discussion of the auto-query feature of Designer 2000 and whether there is any better prospect with Designer 2000 2.0.

The Elements:

Figure A illustrates the module data diagram of the calling form: a simplified order entry screen. 

Figure A 

This illustrates the data diagram of the calling form.

From this form, the user can call the customer details browsing form. This is implemented with a secondary detailed column usage (DCU) of display type ‘Button’. It can be found in HDD_ORDER_HEADERS as ORD_NO at the bottom of the detailed table usage (DTU). The form call is implemented using the PL/SQL block of ORD_NO:

 COMPLEX: std_call_form('b_cst01',     'ordh.cust_id','cst.id');

We will explain each parameter of this procedure later on.

Figure B illustrates the module data diagram of the called form: a simplified customer detail browse screen. There is nothing complicated about the called form. It is a normal one DTU module that can be called either from a menu or from a form, with parameters or without.

Figure B

This illustrates the data diagram of the called form
The Result:

In Figure C, the user is looking at an interesting order placed by our friend Alfred.

Figure C

This illustrates the calling form.
If the user need to verify the address of the customer Alfred, s/he need only click on the press button ‘Customers…’.

This illustrates the called form.
After all, this is a Graphical User Interface. To restrict navigation to the menu is simply not acceptable any more.

On entry in the called form, the details of Alfred were displayed automatically. The user may use this form in any way from then. It can even be queried again. The initial query will not interfere as it is cleared from the parameters after query. 

The Method:

First you must set a few application level preferences for the forms generator. We recommend that you define a named preference set and attach it at application level. The reason for this is that you can then re-use the same preferences for other application systems. It becomes easy to provide corporate standards.

To create a named prefrence set in the preference navigator, simply use menu option set->create. You can then display the set preferences with view->named set windows and proceed to define the values. You can attach a named set of preferences at application level with one click and drag. Click on the label symbol of your new preference set. Then drag it in the preference hierarchy navigator, just underneath the application node.

End user interface preferences:

*Autoqy = N.

This prevents the auto query feature of Designer 2000 (explained at the end of this article) from interfering with what we’ll set up.

*Bschmd  = N

*Bscscp = F

*Bsmoff = A 

*Bsmon = A 

It is our personal preference not to complicate the users’ life with something that they never use anyway. Also this block synchronization feature may interfere if our technique is used on more than one block in the called form. This is possible but is not covered in this article.

Note: you may have to consider performance issues if you design forms with 4 or more blocks.

*Gennav = Y 

This ensures that on entry in the form, the  cursor be placed in the first block of the form with one or more enterable fields. This is important because the automatic query that we will explain bellow needs to happen in a queryable block.

Generate Options preferences:

*Blktgs = After

*Itmtgs = After

By default, these 2 preferences are set to ‘Override’. Our technique will not work if the item and block level triggers override the form level trigger. 

*Callfm = No

This prevents Designer 2000 from generating any “form call form” code to implement the module network that you may define in the module structure diagrammer.

Template  Form

First we need a few form parameters:

context_n1

context_v1

context_n2

context_v2

context_n3

context_v3

context_n4

context_v4

context_n5

context_v5

They are all of data type ‘char’, length 100. A form can be queried on up to five fields. We have never needed more than this. If you do, it is easy to accommodate.

Context_nx is used for the name of a called form field to query. Context_vy is used for the value to use for the query. This value can be a string of characters, a date or a number converted into characters. It always works because we use the built-in ‘copy’ that does not trigger any validation.

Then we need to add two form level triggers to our template form:

*when_new_block_instance:

std_when_new_block_instance;

*pre_query:

std_pre_query;

These two triggers call procedures that belong to our template library. This library is attached to the template form. This is a very powerful technique. We find in many projects that we seldom get enough time to finish the templates before we start developing forms. As a result we often need to improve the template although many forms have already been completed. A lot of those changes can be made just once in the template library, and all forms can pick them up at run time without any re-genaration or compilation. You may encounter a few locking issues when trying to save modifications to this library while other people are running forms based on it. But this is only a minor inconvenience. 

Template  Library

First, we need the call form procedure in Listing A. We have removed a number of comments due to restricted space. We also cut some repetitive code and replaced it with’<< Repeat pattern until>>’.

Listing A


procedure std_call_form (form_name  in varchar2,

                         p_from_1   in varchar2 := null,

                         p_to_1     in varchar2 := null,

                         p_from_2   in varchar2 := null,

                         p_to_2     in varchar2 := null,

                         p_from_3   in varchar2 := null,

                         p_to_3     in varchar2 := null,

                         p_from_4   in varchar2 := null,

                         p_to_4     in varchar2 := null,

                         p_from_5   in varchar2 := null,

                         p_to_5     in varchar2 := null) is

  param_list    ParamList;

  param_name1   varchar2(30) := 'CONTEXT_N1';

  param_value1  varchar2(30) := 'CONTEXT_V1';

  param_name2   varchar2(30) := 'CONTEXT_N2';

  param_value2  varchar2(30) := 'CONTEXT_V2';

  param_name3   varchar2(30) := 'CONTEXT_N3';

  param_value3  varchar2(30) := 'CONTEXT_V3';

  param_name4   varchar2(30) := 'CONTEXT_N4';

  param_value4  varchar2(30) := 'CONTEXT_V4';

  param_name5   varchar2(30) := 'CONTEXT_N5';

  param_value5  varchar2(30) := 'CONTEXT_V5';

  pvalue1
varchar2(50) := null;

  pvalue2
varchar2(50) := null;

  pvalue3
varchar2(50) := null;

  pvalue4
varchar2(50) := null;

  pvalue5
varchar2(50) := null;

BEGIN

  if name_in('system.block_status') != 'NEW' then

    enter; -- Forces validation of current field.

  end if;

  if name_in('system.form_status') = 'CHANGED' then

    msg_alert('Please, save or cancel your changes before '||

              'calling another form.','W',FALSE);

  else

    -- Get the values from the calling form

    if p_from_1 is not null then

      pvalue1 := name_in(p_from_1);

    end if;

    << Repeat pattern until>>

    if p_from_5 is not null then 

      pvalue5 := name_in(p_from_5);

    end if;

    -- Prepare a parameter list for the call_form

    param_list := std_create_Params_List(

                       param_name1, p_to_1,

                       param_value1, pvalue1,

                       << Repeat pattern until>>

                       param_name5, p_to_5,

                       param_value5, pvalue5); 

    CALL_FORM (form_name,NO_HIDE,NO_REPLACE,

               NO_QUERY_ONLY,param_list);

  end if;

END;


Standard call to another form, providing context sensitivity.

The first parameter is the name of the  called form, with no extension. The following parameters go in pairs. The first one is the name of the field in the calling form whose value must be passed as a parameter. The second one is the name of the field in the called form that will receive this value. These two field names must be complete: <block_name>.<field_name>. Example: std_call_form('b_cst01',     'ordh.cust_id' ,'cst.id'); 

The block name can be found in the DTU Table Alias field. The field name is usually the same as the DCU name. If you are not sure or if you get errors, open the generated forms in Developer 2000 and check them out.

Procedure std_call_form uses another procedure whose code is shown in listing B.

Listing B


function std_create_params_list(

                             param_name1 IN varchar2 := NULL,

                             value_for_param1  IN varchar2 := NULL,

                             << Repeat pattern until>>

                             param_name10 IN varchar2 := NULL,

                             value_for_param10  IN varchar2 := NULL)

         return  ParamList is

  param_id  ParamList;

BEGIN

  param_id := Get_Parameter_List('param_id');

  IF NOT ID_NULL(param_id) THEN

    Destroy_Parameter_List(param_id);

  END IF;

  param_id := Create_Parameter_List('param_id');

  IF (param_name1 IS NOT NULL AND value_for_param1 IS NOT NULL) THEN

    Add_Parameter(param_id,param_name1,TEXT_PARAMETER,value_for_param1);

  END IF;

  <<repeat pattern until>>

  IF (param_name10 IS NOT NULL AND value_for_param10 IS NOT NULL) THEN

    Add_Parameter(param_id,param_name10,TEXT_PARAMETER,value_for_param10);

  END IF;

  return (param_id);

END;


Create a parameter list for passing parameters to called forms on startup.

There is nothing complicated about these two procedures. They only prepare parameters and call the required form.

On entry in the called form, the procedure in listing C gets called for each block. This is just a “library hook” that we also use for other template tricks that may be the subject of other articles.

Listing C


procedure std_when_new_block_instance is

begin

 std_context_query;

end;


Library hook.

The fun starts in Listing D. This is where we kick off the query if anything is found in the first parameter. Once the query is executed, we clear the parameters so that they do not interfere with further user querying.

Listing D


procedure std_context_query is

begin

-- If the first parameter is not null, bring up (query) the context

  if name_in('Parameter.context_n1') is not null and

     name_in('Parameter.context_v1') is not null then

    do_key('Execute_Query');

    --

    -- As the query is only required on start-up, set the

    -- parameter to null, once the query has been executed.

    --

    copy(null, 'Parameter.context_n1');

    copy(null, 'Parameter.context_v1');

<<repeat pattern until>>

    copy(null, 'Parameter.context_n5');

    copy(null, 'Parameter.context_v5');

  end if;

end;


Execute a query based on the value held in the form's parameter(s).

The procedure in listing E is run by the form level pre-query trigger. It simply copies parameter values into the relevant fields. It does not matter whether the recipient fields are date fields or number fields as no validation is triggered by the copy built-in.

Listing E


procedure std_pre_query is

  cur_blk       varchar2(30) := Name_In('System.Cursor_Block');

  cur_form      varchar2(30) := Name_In('System.Current_Form');

  Field_name
varchar2(50);

  Query_value
varchar2(50);

BEGIN

  if cur_blk = get_form_property(cur_form,FIRST_NAVIGATION_BLOCK) then

    if name_in('Parameter.context_n1') is not null and

       name_in('Parameter.context_v1') is not null then

      field_name := name_in('Parameter.context_n1');

      query_value := name_in('Parameter.context_v1');

      copy(query_value, field_name);

    end if;

<<repeat pattern until>>

    if name_in('Parameter.context_n5') is not null and

       name_in('Parameter.context_v5') is not null then

      field_name := name_in('Parameter.context_n5');

      query_value := name_in('Parameter.context_v5');

      copy(query_value, field_name);

    end if;

  end if;

end;  


Execute query specified in parameters.
That’s all there is to it. Wasn’t it simple? If you take the time to put this code in your template, all the developers in the team will be able to provide any navigation by a simple  call to std_call_form. This can be done from a press button or a menu option. The initial effort of copying this code into your template will pay for itself within a week.

Auto Query

Designer 2000 also offers the AutoQuery feature. 

We disabled it by setting preference Autoqy to N.

This feature allows you to get a called form to auto query itself depending of the context of the  calling form. However there are a few conditions to get it to work.

· Set the AUTOQY preference to Y for both calling form and called form.

· Define the first block of the called form as a query-only block.

· Base the first block of the called form on a table or view with the same primary key as the relevant block in the calling form.

This works well for drill down, but not for the drill across that we illustrated. We invite you to imagine how you would design the browse customer form so that it is the detail block of a master block based on order number. Good luck to you. 

When we have used the auto-query feature we found that the called forms were using convoluted dummy blocks designs based on views, displayed some unnecessary information and were not re-usable. We also could also not query on non-key fields.

We have inquired about Designer 2000 2.0. It does not provide the drill ‘anywhere’ feature that we need. So we will be using the techique described in this article for a while. And this will provide our users with a quick and user friendly access to their information. This is ultimate purpose any information system.

Hervé Deschamps is a Senior Principal Consultant for Oracle Corporation. He has had five years of experience with Oracle Development Environments (Designer 2000, Developer 2000 and their previous versions). He specialises in tending towards “100% Generation.” You can reach him by e-mail at hdeschamps@sprynet.com. He will provide the code to people who do not like typing. He will be glad to receive any comment or answer any questions.

